


# Pour les fidèles de la TI-57 classique

Nous avons reçu en un an plusieurs dizaines de programmes pour cette petite machine qui fait preuve d'une vitalité remarquable. Faute de place, certaines de ces propositions, souvent intéressantes, n'avaient pas pu donner lieu à un article. Voici pour combler en partie ce retard, une sélection aussi éclectique que possible des meilleures idées que vous nous avez adressées. Les programmes ont été retenus en fonction de leurs qualités propres (bonne programmation), de leur ingéniosité ou de l'originalité du thème abordé.



## Spéculation

■ Votre oncle d'Amérique, dont vous ne connaissiez même pas l'existence, est décédé voilà maintenant un mois. Son testament vient d'être ouvert : il fait de vous son unique héritier ! Il vous lègue des actions cotées en bourse et une certaine somme d'argent liquide.

Cet oncle, Nestor, était un habile

spéculateur. Parti de presque rien, il avait édifié une véritable fortune. Mais, l'héritage n'est pas toujours facile à gérer. Les actions sont soumises aux lois du marché boursier (ici, il est aléatoire) : leur cote peut être en hausse ou en baisse, impossible à prévoir. Selon ce que vous « sentez », il faudra vendre ou acheter... Pourrez-vous ainsi doubler l'héritage de Nestor ou courrez-vous à la ruine ?

□ Sylvain Roques

### Mode d'emploi du programme

On commence avec 0.xxxx STO 5  
 nombre d'actions héritées : A STO 0  
 cote de chaque action : B STO 1  
 argent liquide : C STO 2

On fixe aussi la hausse (ou la baisse) maximum possible à chaque séance, en tapant : D STO 4 (par exemple, si chaque action est cotée 250 francs, on peut fixer le maximum de gain ou de perte à 50 francs, soit B = 250 et D = 50).

Puis on fera RCL 0 × RCL 1 + RCL 2 = STO 3. Ceci donne une estimation de la fortune totale. On gagne lorsque l'on arrive à doubler cette fortune. Pour contrôler ce résultat, on pourra faire RCL 3 × 2 = STO 6, et il suffira de comparer sa fortune avec la mémoire M6 pour savoir si l'on a gagné.

A chaque séance, on a trois possibilités :

1 - ne rien faire : on tape SBR 4  
 2 - vendre n actions : n RST R/S  
 3 - acheter n actions : n SBR 9  
 (si l'affichage clignote, l'opération est impossible ; on tape CLR et on recommence). La machine répond en affichant successivement :

- la fluctuation  
 - le bénéfice (ou la perte) réalisé  
 - la fortune totale disponible (à comparer à M6).

On peut connaître à tout moment :  
 - les actions disponibles à la vente : RCL 0

- le cours actuel des actions : RCL 1  
 - l'argent disponible : RCL 2.

On gagne si l'on arrive à doubler la fortune de départ, c'est-à-dire à atteindre la valeur qui est dans la mémoire M6. On perd par faillite, si la fortune totale disponible devient négative.

### Spéculation

Programme pour TI-57 (à diodes)

Auteur Sylvain Roques

Copyright l'Ordinateur de poche et l'auteur

00	32 7	STO 7	22	22	x ⇌ t
01	33 0	RCL 0	23	34 0	SUM 0
02	-76	2nd INV x ≥ t	24	86 4	2nd Lbl 4
03	51 8	GTO 8	25	33 5	RCL 5
04	33 7	RCL 7	26	-18	2nd INV Log
05	-34 0	INV SUM 0	27	-49	2nd INV Int
06	55	x	28	32 5	STO 5
07	33 1	RCL 1	29	55	x
08	85	=	30	33 4	RCL 4
09	34 2	SUM 2	31	55	x
10	51 4	GTO 4	32	02	2
11	86 9	2nd Lbl 9	33	65	-
12	32 7	STO 7	34	33 4	RCL 4
13	33 1	RCL 1	35	85	=
14	39 7	2nd Prd 7	36	49	2nd Int
15	33 2	RCL 2	37	36	2nd Pause
16	-76	2nd INV x ≥ t	38	34 1	SUM 1
17	51 8	GTO 8	39	33 0	RCL 0
18	33 7	RCL 7	40	55	x
19	-34 2	INV SUM 2	41	33 1	RCL 1
20	33 1	RCL 1	42	75	+
21	-39 7	2nd INV Prd 7	43	33 2	RCL 2
			44	65	-
			45	38 3	2nd Exc 3
			46	85	=
			47	36	2nd Pause
			48	33 3	RCL 3
			49	81	R/S

# Si j'étais goûteur de plats chez Cléopâtre

■ Imaginez un instant que vous êtes un esclave en vente sur un marché de l'Égypte ancienne. Tout à coup, un grand silence... Mais oui, c'est elle ! C'est Cléopâtre en personne qui fait son marché. Et le plus extraordinaire, c'est qu'elle vous choisit. Arrivé au Palais, vous déchantez un peu : vous êtes le

nouveau goûteur des plats de Cléopâtre (le précédent est mort empoisonné).

Vous devez d'abord, parmi quatre plats, en choisir un pour un ennemi de Cléopâtre mais vous ne le goûtez pas. S'il est empoisonné, tout va bien... pour vous. Vous gagnez un point de liberté. Vous n'avez pas à choisir entre les trois plats restants : ils sont tous sains. On vous propose alors quatre autres plats.

Si au contraire vous aviez choisi l'un des plats sains, il en va tout autrement. Non seulement vous n'avez rien gagné, mais cette fois-ci, vous devez goûter vous-même l'un



des trois plats restants. Si ce plat est sain, tant mieux pour vous ; on vous en propose quatre nouveaux. Mais s'il est empoisonné, deux cas se présentent :

- ou vous aviez deux ou trois points de liberté, vous en perdez un et l'on recommence à zéro ;

- ou vous n'aviez plus qu'un seul point de liberté et vous êtes éliminé : 9.999999 99 clignote à l'affichage.

Pour gagner à ce jeu, il faut parvenir à un total de quatre points. Vous êtes alors libre et la machine vous l'indique en affichant un 4 clignotant.

□ Guy Magniant

## Pour les fidèles de la TI-57 classique

### Mode d'emploi du programme

On commence en tapant : 0.xxxx STO 1, 1 STO 0, 1234 STO 4, RST R/S.

On obtient un premier affichage de 1 (vous avez un point de liberté) suivi de 1234 : ce sont les numéros des plats à choisir.

Vous répondez en tapant 1, ou 2, ou 3, ou 4, suivi de R/S, pour choisir le plat empoisonné.

S'il s'agissait effectivement du plat empoisonné, la machine affiche 2 (vous avez gagné un point de liberté), puis 1234 et on recommence.

Si vous avez mal choisi, votre nombre reste affiché, et vous devez en choisir un autre parmi les trois qui restent.

Si vous choisissez encore un plat sain, la machine vous propose à nouveau quatre plats, et on recommence.

S'il était empoisonné, vous perdez un point de liberté. Était-ce le dernier point ? Vous obtenez alors l'affichage clignotant de 9.999999 99 : vous avez perdu.

A l'inverse, si vous parvenez à totaliser 4 points, vous avez gagné : affichage clignotant du chiffre 4.

### Goûteur de plats chez Cléopâtre

Programme pour TI-57 (à diodes)

Auteur Guy Magniant

Copyright l'Ordinateur de poche et l'auteur

00	61 1		SBR 1
01	33 0		RCL 0
02	36	2nd	Pause
03	33 4		RCL 4
04	81		R/S
05	61 2		SBR 2
06	81		R/S
07	51 4		GTO 4
08	81		R/S
09	86 1	2nd	Lbl 1
10	33 1		RCL 1
11	-18	2nd	INV log
12	-49	2nd	INV Int
13	32 1		STO 1
14	55		x
15	04		4
16	75		+
17	01		1
18	85		=
19	49	2nd	Int
20	32 7		STO 7
21	15		CLR
22	-61		INV SBR
23	86 2	2nd	Lbl 2
24	66	2nd	x = t
25	51 3		GTO 3
26	-61		INV SBR
27	86 3	2nd	Lbl 3
28	01		1
29	34 0		SUM 0
30	51 6		GTO 6
31	86 4	2nd	Lbl 4
32	66	2nd	x = t
33	51 5		GTO 5
34	71		RST
35	86 5	2nd	Lbl 5
36	56	2nd	Dsz
37	71		RST
38	45		÷
39	00		0
40	85		=
41	86 6	2nd	Lbl 6
42	04		4
43	22		x ≈ t
44	33 0		RCL 0
45	76	2nd	x ≥ t
46	51 7		GTO 7
47	71		RST

### Exemple d'exécution

Initialisation : 0.4526 STO 1, 1 STO 0, 1234 STO 4 RST, R/S

Affichage	Jeu
1 (pause)	
1234	on fait 4 R/S.
2 (pause)	bon choix : on gagne 1 point.
1234	on fait 2 R/S.
2	on doit choisir un autre plat : on tape 3 R/S.
2 (pause)	le plat 3 était donc sain.
1234	on fait 1 R/S.
1	on doit choisir un autre plat : on tape 4 R/S.
1 (pause)	mauvais choix : on perd 1 point.
1234	on fait 2 R/S.
2	on doit choisir un autre plat : on fait 3 R/S.
9.999999 99	on a perdu.

Pour une autre partie : 1 STO 0, RST, R/S et bonne chance.


# Fantômes et dragons

■ Le jeu se décompose en deux parties. La première consiste (vous vous l'aviez deviné sans doute) à introduire le programme dans la TI-57 ! Comme on utilise la notation alphabétique, cette opération réclame beaucoup d'attention. On pourra vérifier qu'elle s'est bien déroulée en se reportant à la « liste des codes obtenus ». Si certains de ces codes sont différents..., il faut tout recommencer.

Une fois que vous aurez franchi cette petite épreuve, vous pourrez partir à la recherche d'un trésor caché dans un château plutôt étrange. Il est hanté par des dragons qui cherchent à vous éliminer, mais aussi par des fantômes qui vous offriront à l'occasion un délicieux élixir de vie.

Vous saurez toujours où vous en êtes grâce au nombre des points de vie qui vous sont alloués dès le début. S'ils sont négatifs, cela signifie que vous avez été dévoré. Sinon, vous pouvez aller jusqu'au « trésor ».

□ Philippe Morales



## Mode d'emploi

Taper 0.xxxx STO 1  
200 STO 3 (points de vie)  
199 STO 5 (nbre de salles)

On commence en tapant SBR 0, sans regarder l'affichage, puis STO 0 et CLR. On peut alors regarder.

On joue en tapant le déplacement, suivi de SBR 0.

### Exemple :

50 SBR 0 : on va en salle 50  
affichage F : un fantôme  
affichage 265 : vos points de vie  
affichage -50 clignotant : le trésor est plus haut  
On fait CLR.

12 SBR 0 : on va en salle 62 (soit 50 + 12)  
affichage d : un dragon  
affichage 189 : vos points de vie  
affichage 62 clignotant : le trésor est plus bas  
On fait CLR.

5 +/- SBR 0 : on va en salle 57  
affichage d : encore un dragon  
affichage 123 : vos points de vie  
affichage TRESOR  
puis 123  
puis TRESOR  
puis 123

Bravo : vous avez gagné (c'est en général plus long !).

## Liste des codes obtenus

00	07	25	61 6
01	10	26	76
02	14	27	51 1
03	05	28	00
04	00	29	15
05	10	30	36
06	36	31	61 6
07	51 9	32	34 3
08	86 6	33	51 9
09	33 1	34	86 1
10	-18	35	00
11	-49	36	13
12	32 1	37	36
13	55	38	61 6
14	33 5	39	-34 3
15	75	40	86 9
16	01	41	33 0
17	85	42	32 7
18	49	43	33 3
19	-61	44	36
20	86 0	45	33 2
21	34 2	46	66
22	-34 3	47	71
23	61 6	48	-76
24	32 7	49	84

## Fantômes et dragons

Programme pour TI-57 (à diodes)  
Auteur Philippe Morales  
Copyright l'Ordinateur de poche et l'auteur

Le programme comprenant un affichage alphabétique, les pas n'ont pas été numérotés, il est impératif de suivre exactement la programmation indiquée (un chiffre entre parenthèses indique la répétition de la séquence : ainsi SST (2) signifie SST, SST).

LRN		LRN	RST
2nd	Exc		R/S (2)
	SST		RST
2nd	Lbl 1	LRN	
	0	2nd	Del (4)
	R/S		SST (6)
	7	2nd	Ins (23)
	2	2nd	Pause
	6		GTO 9
	5	2nd	Lbl 6
	0		RCL 1
	6	2nd	INV log
	7	2nd	INV Int
	5		STO 1
	RST		x
LRN			RCL 5
	RST		+
	R/S (2)		1
	SST		=
LRN		2nd	Int
	2		INV SBR
LRN		2nd	Lbl 0
	RST		SUM 2
	R/S (2)		INV SUM 3
	SST (2)		SBR 6
LRN			STO 7
	6		SBR 6
LRN		2nd	x ≥ t
	RST		GTO 1
	R/S (2)		SST (2)
	SST (3)	2nd	Ins (6)
LRN		2nd	Pause
LRN			SBR 6
	RST		SUM 3
	R/S (2)		GTO 9
	SST (5)	2nd	Lbl 1
LRN			SST (2)
	2	2nd	Pause
LRN			SBR 6
	RST		INV SUM 3
	R/S (2)	2nd	Lbl 9
	SST (6)		RCL 0
LRN			STO 7
	7		RCL 3
LRN		2nd	Pause
	RST		RCL 2
	R/S (2)	2nd	x = t
	SST (7)		RST
LRN		2nd	INV x = t
	5		+/-

# Du brouillage sur les ondes

## Pour les fidèles de la TI-57 classique


■ La TI-57 émet, quand elle est en marche et spécialement lorsqu'elle n'est pas branchée sur le secteur, un léger bruit que l'on peut discerner en y prêtant une oreille attentive.

Si l'on approche la machine d'un poste de radio réglé sur un emplacement non occupé des grandes ondes, on obtient un bruissement continu. Le récepteur radio sert alors d'amplificateur, et la TI-57, placée à deux centimètres du poste ou plus près encore, se comporte

comme un émetteur. Il ne reste plus qu'à inventer des programmes dont l'exécution produit des effets de bruitage.

Selon la façon dont on place la TI-57, on modifie sensiblement le timbre des sons émis. Il vous revient donc de rechercher la disposition produisant les sonorités qui vous conviennent le mieux. On trouvera ci-dessous quelques exemples de ces programmes bruyants.

□ Hugues Biratelle



### Bruitages

Programmes pour TI-57 (à diodes)  
Auteur Hugues Biratelle  
Copyright l'Ordinateur de poche et l'auteur

#### A l'écoute des ondes

00		1
01		SUM 0
02		RCL 0
03	2nd	x = t
04		GTO 1
05		RCL 1
06		RST
07	2nd	Lbl 1
08		RCL 1
09		RCL 2
10		RCL 3
11		RCL 4
12		RCL 5
13		RCL 6
14		RCL 5
15		RCL 4
16		RCL 3
17		RCL 2
18		RCL 1
19		GTO 1

On entre : 9.9 STO 1  
9.99 STO 2  
9.999 STO 3  
9.9999 STO 4  
9.99999 STO 5  
9.999999 STO 6  
10 STO 7

Puis on fait RST 0 STO 0, et R/S.  
Un navigateur essaie de capter une radio-balise. Au début le signal est très faible, puis tout à coup il est reçu 5 sur 5.

#### Avion à hélices

00	9
01	RST

On fait RST suivi de R/S pour obtenir le bruit d'un avion à hélices.

#### Imprimerie

00	RCL 0
01	1/x
02	RST

Faire 99999999 STO 0, puis RST et R/S.

#### Teuf-teuf...

00	1
01	SUM 2
02	RCL 2
03	EE
04	0
05	1
06	2
07	3
08	4
09	5
10	6
11	7
12	8
13	9
14	8
15	7
16	6
17	5
18	4
19	3
20	2
21	1
22	0
23	RST

On fait RST, suivi de R/S et le vieux tacot démarre aussitôt.

#### Hélicoptère

00	RCL 0
01	RCL 1
02	RCL 2
03	RCL 3
04	RCL 4
05	RCL 5
06	RCL 6
07	RCL 7
08	RCL 7
09	RCL 7
10	RCL 7
11	RCL 7
12	RCL 6
13	RCL 5
14	RCL 4
15	RCL 3
16	RCL 2
17	RCL 1
18	RCL 0
19	RST

Faire 1 STO 0  
11 STO 1  
111 STO 2  
1111 STO 3  
11111 STO 4  
111111 STO 5  
1111111 STO 6  
11111111 STO 7

Puis RST et R/S ; on obtient alors sur la radio le bruit des pales d'un hélicoptère.

#### Cheval au galop

00	1
01	SUM 2
02	RCL 2
03	EE
04	1
05	SUM 2
06	RCL 2
07	RST

On fait simplement RST suivi de R/S, et le cheval part au galop.

# Résoudre une équation du second degré en 23 pas

**Equations du second degré**  
Programme pour TI-57 (à diodes)  
Auteur Eric Wolf  
Copyright l'Ordinateur de poche  
et l'auteur

00	25	1/x
01	55	x
02	22	$x \approx t$
03	81	R/S
04	45	÷
05	02	2
06	85	=
07	84	+/-
08	32 0	STO 0
09	23	$x^2$
10	65	-
11	33 0	RCL 0
12	22	$x \approx t$
13	55	x
14	81	R/S
15	85	=
16	24	$\sqrt{x}$
17	65	-
18	22	$x \approx t$
19	34 7	SUM 7
20	85	=
21	84	+/-
22	81	R/S

■ Les coefficients sont introduits dans l'ordre attendu : a, b puis c, et la TI-57 fait le reste. Elle affiche donc les deux racines de l'équation. Si la racine est complexe, la partie imaginaire est donnée en premier (l'affichage est alors clignotant) et l'on presse sur la touche  $x \approx t$  pour obtenir la partie réelle. C'est aussi simple que cela.

□ Eric Wolf

## 1<sup>er</sup> exemple d'utilisation

Soit l'équation  $x^2 - 2x - 3 = 0$ . Faire RST, puis taper 1 R/S, 2 +/- R/S, 3 +/- R/S. On obtient un affichage de -1 qui représente la première racine. On fait alors  $x \approx t$  pour obtenir 3 qui est la seconde racine.

## 2<sup>e</sup> exemple d'utilisation

Si la racine est complexe, la machine indiquera d'abord la partie imaginaire au moyen d'un affichage clignotant. On pressera la touche  $x \approx t$  pour connaître la partie réelle. Ainsi pour  $x^2 + 2x + 17 = 0$ , on tapera RST, 1, R/S, 2, R/S, 17 et R/S. L'affichage clignote sur 4, partie imaginaire. Une pression sur  $x \approx t$  et l'on obtient -1 qui est la partie réelle. On a donc  $x_1 = -1 + 4i$  et  $x_2 = -1 - 4i$ .

# Une chasse aux papillons

■ Les papillons volettent de fleur en fleur. Vous voudriez bien en attraper quelques-uns pour parfaire votre collection. Mais ce n'est pas si facile. Pour abattre le filet, il faut attendre que le papillon se place sur une fleur, au centre d'un bouquet de cinq. Si c'était le bon moment, vous augmentez votre collection d'une unité. Sinon, vous perdez un

papillon : il s'est échappé. Au début de la chasse, vous en déclarez un certain nombre, une sorte de provision. Dès que ce nombre devient négatif (tous les papillons se sont échappés), la chasse prend fin.

Sinon, vous pouvez augmenter votre collection... indéfiniment.

□ François Fayard

## Chasse aux papillons

Programme pour TI-57 (à diodes)  
Auteur François Fayard  
Copyright l'Ordinateur de poche  
et l'auteur

00	33 6	RCL 6
01	36	2nd Pause
02	75	+
03	33 7	RCL 7
04	85	=
05	-76	2nd INV $x \geq t$
06	51 7	GTO 7
07	09	9
08	32 0	STO 0
09	86 1	2nd Lbl 1
10	-56	2nd INV Dsz
11	51 0	GTO 0
12	33 3	RCL 3
13	-56	2nd INV Dsz
14	51 0	GTO 0
15	33 1	RCL 1
16	-56	2nd INV Dsz
17	51 0	GTO 0
18	33 2	RCL 2
19	-56	2nd INV Dsz
20	51 0	GTO 0

21	38 1	2nd Exc 1
22	-56	2nd INV Dsz
23	51 0	GTO 0
24	38 3	2nd Exc 3
25	-56	2nd INV Dsz
26	51 0	GTO 0
27	38 2	2nd Exc 2
28	32 1	STO 1
29	51 1	GTO 1
30	86 0	2nd Lbl 0
31	01	1
32	-34 6	INV SUM 6
33	51 5	GTO 5
34	86 4	2nd Lbl 4
35	-66	2nd INV $x = t$
36	51 0	GTO 0
37	01	1
38	34 6	SUM 6
39	86 5	2nd Lbl 5
40	33 4	RCL 4
41	32 1	STO 1
42	33 7	RCL 7
43	32 2	STO 2
44	33 5	RCL 5
45	32 3	STO 3
46	71	RST

## Exemple d'exécution

Les fleurs sont représentées par des « 1 », les papillons par des « 8 ». On abat le filet en faisant R/S, puis SBR 4.

En début de partie, taper dans l'ordre : 81111 STO 1 STO 4, 11811 STO 2 STO 7 et 11118 STO 3 STO 5. Introduire alors la « provision » de papillons en mémoire 6. Par exemple : 5 STO 6 (à chaque nouvelle partie, il faudra remettre un nombre dans cette mémoire 6).

On lance le programme par RST puis R/S. A l'affichage, le papillon voleté de fleur en fleur : 81111, 11811, 81111, 11118, 11811. Le filet sera abattu à ce moment-là par R/S puis SBR 4. Avant de rabattre le filet, il faut attendre le résultat : 6, un papillon a été attrapé.

Si vous attendez trop longtemps avant d'agir, vous obtenez le même résultat que si vous agissiez au mauvais moment : vous laissez un papillon s'échapper.